Derivadas

1 [redacted]fica de la hipérbola xy=1 y calcula los puntos de la gráfica de dicha hipérbola en los que la recta tangente pasa por el punto (–1,1).

Sabiendo que la recta tangente de una función es g(x)=f(a)+f′(a)(x–a):

```
define(f(x),1/x);
define(df(x), diff(f(x),x));
define(g(x,a), f(a)+df(a)·(x–a));
```

$$f(x):=\frac{1}{x}$$

$$df(x):=-\frac{1}{x^2}$$

$$g(x,a):=\frac{1}{a}-\frac{x-a}{a^2}$$

Sustituimos y resolvemos con los puntos x,y dados (x=-1, y=1).

```
solve(g(–1,a)=1,a);
```

$$[a=1-\sqrt{2},a=\sqrt{2}+1]$$

```
wxdraw2d(
   explicit(f(x), x, −10, 10),
   color=red,
   explicit(g(x,1−sqrt(2)),x,−10,10),
   color=pink,
   explicit(g(x,sqrt(2)+1),x,−10,10),
   yrange= [−10,10],
   color=black,
   point_type=7,
   point_size=2,
   points([
               [1−sqrt(2), f(1−sqrt(2))],
               [sqrt(2)+1,   f(sqrt(2)+1)]
            ]),
   color=red,
    points([
               [−1, 1]
            ]),
   yaxis=true,
   xaxis=true,
   grid=true,
   proportional_axes=xy
);
```



2 Haz la gráfica de la función f(x)=2x^3+13x^2+5x+9 y calcula los puntos de la gráfica en los que la recta tangente pasa por el origen de coordenadas.

Mismo procedimiento que antes para sacar la recta tangente:

```
define(f(x),2·x^3+13·x^2+5·x+9);
define(df(x), diff(f(x),x));
define(g(x,a), f(a)+df(a)·(x−a));
```

$$f(x):=2\,x^3+13\,x^2+5\,x+9$$

$$\mathrm{df}(x):=6\,x^2+26\,x+5$$

$$\mathrm{g}(x,a):=(6\,a^2+26\,a+5)\,(x-a)+2\,a^3+13\,a^2+5\,a+9$$

Sustituimos y resolvemos con los puntos x,y dados (x=0, y=0).

```
solve(g(0,a)=0,a);
```

$$\left[a=\frac{3}{4},a=-1,a=-3\right]$$

```
wxdraw2d(
   explicit(f(x), x, -7, 5),
   color=red,
   explicit(g(x,3/4),x,-7,5),
   color=pink,
   explicit(g(x,-1),x,-7,5),
   color=purple,
   explicit(g(x,-3),x,-7,5),
   yrange= [-5,60],
   color=black,
   point_type=7,
   point_size=2,
   points([
               [3/4, f(3/4)],
               [-1, f(-1)],
               [-3, f(-3)]
            ]),
   color=red,
    points([
               [-0, 0]
            ]),
   yaxis=true,
   xaxis=true,
   grid=true
);
```



3 Dibuja la semicircunferencia superior centrada en el origen y de radio 2. Dibuja el punto en el que la recta tangente a dicha circunferencia pasa por el punto (3,7), así como dicha recta tangente.

Sabiendo que la ecuación de una circunferencia es $x^2+y^2=r^2$, y=√x²-4. Mismo procedimiento que en los anteriores:

```
define(f(x),sqrt(4–(x^2)));
define(df(x),diff(f(x),x));
define(g(x,a), f(a)+df(a)·(x–a));
```

$$f(x):=\sqrt{4-x^2}$$

$$df(x):=-\frac{x}{\sqrt{4-x^2}}$$

$$g(x,a):=\sqrt{4-a^2}-\frac{a\,(x-a)}{\sqrt{4-a^2}}$$

A maxima cuesta resolver g(x,a), así que vamos a "dejarla bonita": es equivalente a (4-a*x)* sqrt((2-a)*(2+a)) /((2-a)*(2+a)).

```
t(x,a):= (4–a·x)· sqrt((2–a)·(2+a)) /((2–a)·(2+a));
```

$$t(x,a):=\frac{(4-a\,x)\sqrt{(2-a)\,(2+a)}}{(2-a)\,(2+a)}$$

```
to_poly_solve(t(3,a)=7,a);
```

WARNING: redefining MAXIMA::OPAPPLY in DEFMACRO
WARNING: redefining MAXIMA::OPCONS in DEFMACRO

$$\%union\left(\left[a=-\frac{21\sqrt{6}-6}{29}\right]\right)$$

```
wxdraw2d(
   implicit(x²+y²=4, x, −10, 10, y, −10, 10),
   color=red,
   explicit(g(x,−(21·sqrt(6)−6)/29),x,−10,10),
   yrange= [−10,10],
   color=black,
   point_type=7,
   point_size=2,
   points([
               [−(21·sqrt(6)−6)/29, f(−(21·sqrt(6)−6)/29)]
            ]),
   color=red,
    points([
               [3, 7]
            ]),
   yaxis=true,
   xaxis=true,
   grid=true,
   proportional_axes=xy
);
```



4 Calcula la imagen de la función f de R en R definida por f(x)=e−x2(x2−3)

```
define(f(x),%e^(−x²)·(x²−3));
define(df(x),diff(f(x),x));
```

$$f(x):=(x^2-3)\,\%e^{-x^2}$$

$$df(x):=2\,x\,\%e^{-x^2}-2\,x\,(x^2-3)\,\%e^{-x^2}$$

Buscamos los puntos críticos de la función:

```
solve(df(x)=0);
```

$$[x=-2, x=2, x=0]$$

Y ahora vemos los intervalos de monotonía en los intervalos:
[-∞,-2][-2,0][0,2][2,∞]

```
sign(df(−10));
sign(df(−1));
sign(df(1));
sign(df(10));
```

pos
neg
pos
neg

[-∞,-2] -> creciente
[-2,0] -> decreciente
[0,2] -> creciente
[2,∞] -> decreciente

Por esto concluimos que los puntos 2,-2 son máximos relativos y el punto 0 es mínimo relativo. Podemos comprobarlo en la gráfica:

```
wxdraw2d(
   explicit(f(x),x,−3,3),
   yrange=[−3,1],
   yaxis=true,
   xaxis=true,
   grid=true
);
```



Para obtener la imagen nos quedaría calcular la unión de f([-∞,-2]) U f([-2,0]), y como es par podemos ahorrarnos el resto. Queda calcular el límite cuando x tiende a infinito:

```
limit(f(x),x,minf);
```

0

```
f(−2);
f(0);
```

$\%e^{-4}$
-3

Entonces, la imagen es:
f([-∞,-2]) ∪ f([-2,0]) =
[0,f(-2)] ∪ [f(-2),f(0)] =
[0, e^(-4)] ∪ [%e^(-4),-3] =

[-3,%e^(-4)]

5 Calcula la imagen de la función definida en los reales positivos por f(x)=x1/x

```
define(f(x),x^(1/x));
define(df(x),diff(f(x),x));
```

$$f(x):=x^{1/x}$$

$$df(x):=x^{1/x}\left(\frac{1}{x^2}-\frac{\log(x)}{x^2}\right)$$

Buscamos los puntos críticos de la función:

```
solve(df(x)=0);
```

$$\left[x=\%e,\frac{1}{x^{\frac{2x-1}{x}}}=0\right]$$

Y ahora vemos los intervalos de monotonía en los intervalos:
[1,e][e,∞]

```
sign(df(1));
sign(df(10));
```

pos
neg

[-∞,e] -> creciente
[e,∞] -> decreciente

Por esto concluimos que el punto e es máximo relativo. Podemos comprobar la gráfica:

```
wxdraw2d(
   explicit(f(x),x,0,15),
   yrange=[-5,5],
   yaxis=true,
   xaxis=true,
   grid=true
);
```



Para obtener la imagen nos queda observar el comportamiento en el 0 por la derecha y en infinito:

```
f(%e);
```

$$\%e^{\%e^{-1}}$$

```
limit(f(x),x,0,plus);
limit(f(x),x,inf);
```

0

1

Entonces, la imagen es:
f([0,%e]) ∪ f([%e,∞]) =
[lim->0,f(%e)] ∪ [f(%e),lim->∞] =
[0,%e^%e^(-1)] ∪ [%e^%e^(-1),1]=

[0,%e^%e^(-1)]

6 Sea f definida en los números reales positivos como f(x)=(x+ex)(1/x). Estudia cómo se podría extender la función en el punto 0 para que fuera continua dicho punto, y si para ese valor es derivable en 0

```
define(f(x),(x+%e^x)^(1/x));
define(df(x),diff(f(x),x));
```

$$f(x):=(\%e^{x}+x)^{1/x}$$

$$df(x):=(\%e^{x}+x)^{1/x}\left(\frac{\%e^{x}+1}{x(\%e^{x}+x)}-\frac{\log(\%e^{x}+x)}{x^{2}}\right)$$

Calculamos los límites de 0 por la derecha y por la izquierda:

```
limit(f(x),x,0,plus);
limit(f(x),x,0,minus);
```

$$\%e^2$$

$$\%e^2$$

Los límites laterales coinciden, por lo cual la función puede extenderse a 0: bastaría con declararlo como una función por partes:
f(x)=(x+%e^x)^(1/x) si x!=0
f(x)=%e^2 si x=0

Para ver si es también derivable, las derivadas en el punto deben coincidir entre sí y con los límites laterales:

```
df(limit(f(x),x,0,plus));
df(limit(f(x),x,0,minus));
```

$$\left(\%e^{\%e^2}+\%e^2\right)^{\%e^{-2}}\left(\frac{\%e^{-2}\left(\%e^{\%e^2}+1\right)}{\%e^{\%e^2}+\%e^2}-\%e^{-4}\log\left(\%e^{\%e^2}+\%e^2\right)\right)$$

$$\left(\%e^{\%e^2}+\%e^2\right)^{\%e^{-2}}\left(\frac{\%e^{-2}\left(\%e^{\%e^2}+1\right)}{\%e^{\%e^2}+\%e^2}-\%e^{-4}\log\left(\%e^{\%e^2}+\%e^2\right)\right)$$

Las derivadas coinciden entre sí, pero no con el límite, por lo cual puede extenderse a 0 pero no es derivable en ese punto.

7 *Calcula la imagen de la función como f(x)=x+e−x−2*

```
define(f(x),x+%e^(−x)−2);
define(df(x),diff(f(x),x));
```

$$f(x):=\%e^{-x}+x-2$$

$$df(x):=1-\%e^{-x}$$

Buscamos los puntos críticos de la función:

```
solve(df(x)=0);
```

$$[x=0]$$

Y ahora vemos los intervalos de monotonía en los intervalos:
[-∞,0][0,∞]

```
sign(df(−1));
sign(df(1));
```

neg
pos

[-∞,e] -> decreciente
[e,∞] -> creciente

Por esto concluimos que el punto 0 es mínimo relativo y absoluto.
Podemos comprobar la gráfica:

```
wxdraw2d(
   explicit(f(x),x,−15,15),
   yrange=[−5,5],
   yaxis=true,
   xaxis=true,
   grid=true
);
```



Para obtener la imagen nos queda observar el comportamiento en más y menos infinito:

```
f(0);
```

-1

```
limit(f(x),x,minf);
limit(f(x),x,inf);
```

∞

∞

Entonces, la imagen es:
f([-∞,0]) ∪ f([0,∞]) =
[lim->-∞,f(0)] ∪ [f(0),lim->∞] =
[∞,-1] ∪ [-1,∞]=

[-1,∞]

8 Calcula las dimensiones de la cruz simétrica respecto de los ejes y con área máxima que se puede inscribir en una circunferencia de radio 1

Si la cruz es simétrica respecto a los ejes (y no simétrica respecto a x=y, por ejemplo (X)), significa que tenemos una cruz en forma de +. Se puede consultar una explicación detallada de este ejercicio en la relación de derivadas, ejercicio 33.

El área de esta figura será 2*(x*y/2) [Los dos rectángulos que conforman la cruz] menos x*(y-x)/2 (el área en la que se superponen).
A = 2*(x*f(x)/2) - (x*(f(x)-x))/2;
A = x*f(x) - x*(f(x)-x)

```
f(x):=sqrt(1−x^2);
```

$$f(x):=\sqrt{1-x^2}$$

La función a optimizar es el área:

```
define(A(x), (x·f(x)) − ((x·(f(x)−x))/2));
define(dA(x),diff(A(x),x));
```

$$A(x):=x\sqrt{1-x^2}-\frac{x\left(\sqrt{1-x^2}-x\right)}{2}$$

$$dA(x):=-\frac{\sqrt{1-x^2}-x}{2}+\sqrt{1-x^2}-\frac{x\left(-\frac{x}{\sqrt{1-x^2}}-1\right)}{2}-\frac{x^2}{\sqrt{1-x^2}}$$

Vamos a ver puntos críticos:

```
to_poly_solve(dA(x)=0,x);
```

WARNING: redefining MAXIMA::OPAPPLY in DEFMACRO
WARNING: redefining MAXIMA::OPCONS in DEFMACRO

$$\%union\left(\left[x=-\frac{\sqrt{2-\sqrt{2}}}{2}\right],\left[x=\frac{\sqrt{\sqrt{2}+2}}{2}\right]\right)$$

Descartamos la opción negativa ya que es simétrica y lo podemos sacar con la opción positiva en el primer cuadrante. Calculamos la y correspondiente a esa x:

```
p_x:sqrt(sqrt(2)+2)/2;
p_y:f(p_x);
```

$$\frac{\sqrt{\sqrt{2}+2}}{2}$$

$$\sqrt{1-\frac{\sqrt{2}+2}{4}}$$

Podemos esbozar la cruz con la orden rectangle:

```
wxdraw2d(
  implicit(x²+y²=1, x, −2, 2, y, −2, 2),
  yrange= [−2,2],
  color=red,
  rectangle([p_x,p_y],[−p_x,−p_y]),
  rectangle([p_y,p_x],[−p_y,−p_x]),
  yaxis=true,
  xaxis=true,
  grid=true,
  proportional_axes=xy
);
```



9 Esboza la gráfica de la función f(x)=1/(x+2)^2 para x>0. Como consecuencia, determina las dimensiones del rectángulo de mayor área que se puede construir apoyado sobre el eje OX y con vértices opuestos en el origen y en un punto de la gráfica anterior.

En este caso nos piden que optimicemos de nuevo el área de un rectángulo (función a optimizar: Área=base*altura/2). Como requisito, la esquina superior derecha del rectángulo debe pertenecer a f(x), así que la función área sería como sigue:

```
define(f(x),1/(x−2)^2);
```

$$f(x):=\frac{1}{(x-2)^2}$$

```
define(A(x),(x·(f(x))/2));
define(dA(x),diff(A(x),x));
```

$$A(x):=\frac{x}{2\,(x-2)^2}$$

$$\mathrm{dA}(x) := \frac{1}{2\,(x-2)^2} - \frac{x}{(x-2)^3}$$

Así que calculamos el maximo de la función área viendo donde se anula la derivada:

```
to_poly_solve(dA(x)=0,x);
```

$$\%\mathrm{union}([x=-2])$$

Aunque nos de una x negativa es correcto, ya que no se especifica el cuadrante, sólo que esté apoyada en el eje x y que su extremo opuesto sea el origen de coordenadas. Graficamos el rectángulo y la f(x):

```
wxdraw2d(
   explicit(f(x), x, −4, 1),
   yrange= [−2,2],
   color=red,
   rectangle([0,0],[−2,f(−2)]),
   yaxis=true,
   xaxis=true,
   grid=true,
   proportional_axes=xy
);
```



Polinomio de Taylor

1 ~~Calcula el polinomio de Ta~~ylor de la función arcotangente en cero de grado 5. Representa gráficamente ambas funciones

```
define(f(x),atan(x));
define(t(x),ratdisrep(taylor(f(x),x,0,5)));
```

$$\mathrm{f}(x) := \mathrm{atan}(x)$$

$$\mathrm{t}(x) := \frac{x^5}{5} - \frac{x^3}{3} + x$$

```
wxdraw2d(
   color=blue,
   explicit(f(x), x, -4, 4),
   color=red,
   explicit(t(x), x, -4, 4),
   yrange= [-4,4],
   yaxis=true,
   xaxis=true,
   grid=true,
   proportional_axes=xy
);
```



2 Calcula el los seis primeros polinomios de Taylor centrados en cero de la función $(x^3-1)^2$

```
define(f(x),(x^3-1)^2);
define(t1(x),ratdisrep(taylor(f(x),x,0,1)));
define(t2(x),ratdisrep(taylor(f(x),x,0,2)));
define(t3(x),ratdisrep(taylor(f(x),x,0,3)));
define(t4(x),ratdisrep(taylor(f(x),x,0,4)));
define(t5(x),ratdisrep(taylor(f(x),x,0,5)));
define(t6(x),ratdisrep(taylor(f(x),x,0,6)));
```

$$\mathrm{f}(x):=(x^3-1)^2$$

$$\mathrm{t1}(x):=1$$

$$\mathrm{t2}(x):=1$$

$$\mathrm{t3}(x):=1-2x^3$$

$$\mathrm{t4}(x):=1-2x^3$$

$$\mathrm{t5}(x):=1-2x^3$$

$$\mathrm{t6}(x):=x^6-2x^3+1$$

```
wxdraw2d(
   color=blue,
   explicit(f(x), x, −4, 4),

   color=red,
   explicit(t1(x), x, −4, 4),
   color=green,
   explicit(t2(x), x, −4, 4),
   color=yellow,
   explicit(t3(x), x, −4, 4),
   color=orange,
   explicit(t4(x), x, −4, 4),
   color=pink,
   explicit(t5(x), x, −4, 4),
   color=black,
   explicit(t6(x), x, −4, 4),

   yrange= [−1,4],
   yaxis=true,
   xaxis=true,
   grid=true,
   proportional_axes=xy
);
```



3 Consideremos la función f(x)=e^(sin(x)+cos(x)) definida en los números reales. En este ejercicio vamos a ver los problemas que pueden acaecer cuando trabajamos con polinomios de Taylor sin aplicarles el comando ratdisrep.

3.1 Define la función f(x)

```
define(f(x),%e^(sin(x)+cos(x)));
```

$$f(x) := \%e^{\sin(x)+\cos(x)}$$

3.2 Define su polinomio de Taylor centrado en 0 de grado 7

```
define(t(x),taylor(f(x),x,0,7));
```

$$t(x):=\%e+\%e\,x-\frac{\%e\,x^3}{2}-\frac{5\,\%e\,x^4}{24}+\frac{\%e\,x^5}{10}+\frac{71\,\%e\,x^6}{720}+\frac{\%e\,x^7}{720}+...$$

3.3 Calcula el error que se comete cuando se toma como valor aproximado de f(1) el valor de el polinomio de Taylor que has calculado...

3.3.1 Evaluando f en 1 y restando el valor del polinomio en 1

```
ratprint: false$
numer:true$
f(1)-t(1);
```

$$-\frac{2645302989515}{36329664276829}$$

```
float(%);
```

$$-0.07281385727536628$$

3.3.2 Definiendo una nueva función como la resta de la función f y su polinomio de Taylor y evaluando dicha función en 1

```
define(error_taylor(x),f(x)-t(x));
```

$$error\_taylor(x):=-\frac{1}{1394659498655744}-\frac{x}{1394659498655744}+\frac{x^3}{2789318997311488}-\frac{x^4}{4339508656996351}-\frac{x^5}{3034084100014080}+\frac{x^6}{9626577335746560}+\frac{x^7}{483567158710763520}+...$$

```
error_taylor(1);
```

$$-\frac{1}{653759755832110}$$

```
float(%);
```

$$-1.529613885038232\ 10^{-15}$$

Vemos que los errores obtenidos son diferentes dependiendo de cómo calculamos el error.

3.4 Comprueba que si al polinomio de Taylor le aplicas la orden ratdisdep, dicha anomalía desaparece.

```
define(f(x),%e^(sin(x)+cos(x)));
define(t(x),ratdisrep(taylor(f(x),x,0,7)));
```

$$f(x):=\%e^{\sin(x)+\cos(x)}$$

$$t(x):=0.003775391428415338\ x^7+0.2680527914174891\ x^6+0.2718281828459049\ x^5-0.5663087142623009\ x^4-1.359140914229523\ x^3+2.718281828459045\ x+2.718281828459045$$

```
f(1)-t(1);
```

$$-0.0728138572753605$$

```
define(error_taylor(x),f(x)-t(x));
```

$$error\_taylor(x):=\%e^{\sin(x)+\cos(x)}-0.003775391428415338\ x^7-0.2680527914174891\ x^6-0.2718281828459049\ x^5+0.5663087142623009\ x^4+1.359140914229523\ x^3-2.718281828459045\ x-2.718281828459045$$

```
error_taylor(1);
```

$$-0.0728138572753605$$

Cuando definimos taylor con ratdisrep el error es el mismo, sin importar cómo lo calculamos.

Integrales y primitivas

1 [redacted]

```
define(f(x),x^3·sin(2·x));
integrate(f(x),x);
```

$$f(x):=x^3\sin(2\ x)$$

$$\frac{(6\ x^2-3)\sin(2\ x)+(6\ x-4\ x^3)\cos(2\ x)}{8}$$

2 Calcula ∫1/(x√1−x^2) dx*

```
define(f(x),1/(x·sqrt(1−x^2)));
integrate(f(x),x);
```

$$f(x):=\frac{1}{x\,(1-x^2)^{0.5}}$$

$$-1.0\,\log\left(\frac{2\sqrt{1-x^2}}{|x|}+\frac{2}{|x|}\right)$$

3 Calcula las derivadas de la función $f(x)=\int[x^2,0]$ sin(log(1+t)) dt

Seguimos el teorema fundamental del cálculo:

```
define(f(t),sin(ln(1+t)));
define(g(x),x^2);
define(h(x),0);
define(dg(x),diff(g(x),x));
define(dh(x),diff(h(x),x));
```

$$f(t):=\sin(\ln(t+1))$$

$$g(x):=x^2$$

$$h(x):=0$$

$$dg(x):=2\,x$$

$$dh(x):=0$$

```
define(df(x),f(g(x))·dg(x) − f(h(x))·dh(x));
```

$$df(x):=2\,x\sin(\ln(x^2+1))$$

4 Calcula el máximo de la función f:[1,∞[→R definida por f(x)=∫[x-1,0] (e^(−t^2) − e^(−2t)) dt

De nuevo hay que aplicar teorema fundamental del cálculo para obtener la derivada de una integral:

```
define(f(t),(%e^(−t^2) − %e^(−2·t)) );
define(g(x),x−1);
define(h(x),0);
define(dg(x),diff(g(x),x));
define(dh(x),diff(h(x),x));
```

$$f(t):=\%e^{-t^2}-\%e^{-2\,t}$$

$$g(x):=x-1$$

$$h(x):=0$$

$$dg(x):=1$$

$$dh(x) := 0$$

```
define(df(x),f(g(x))·dg(x) − f(h(x))·dh(x));
```

$$df(x) := \%e^{-(x-1)^2} - \%e^{-2(x-1)}$$

Para encontrar el máximo, vemos donde se anula la derivada. Primero la graficamos para ver dónde cambia de signo e intentar usar find_root:

```
wxplot2d([df(x)], [x,−300,300])$
```



```
is(df(5000)>0);
```

false

Como vemos, la derivada no se anula nunca, así que no podemos resolverla con find_root ni con ninguna ecuación. La función no tiene máximo porque tiene una asíntota horizontal en 0.

5 Calcula lim x→0 (x * [∫[x,0] sin(t^2) dt]) / sin(x^4)

Maxima nos pregunta por la x. Como vamos a calcular el límite a 0, vamos a hacer 3 versiones de la función para ver el límite en 0, por la derecha y por la izquierda.

```
define(f_plus(x), (x · integrate(sin(t^2),t,0,x))/sin(x^4));
```

*Is x positive, negative or zero?*positive;

$$f\_plus(x) := 6.236666813542231\,10^{-16}\,x\,((251198712145819\,\%i + 251198712145819)\,\mathrm{erf}(3.126088660375976\,10^{-8}\,(22619537\,\%i + 22619537)\,x) + (251198712145819\,\%i - 251198712145819)\,\mathrm{erf}(3.126088660375976\,10^{-8}\,(22619537\,\%i - 22619537)\,x) + (251198712145819 - 251198712145819\,\%i)\,\mathrm{erf}((-\%i)^{0.5}\,x) + (251198712145819\,\%i + 251198712145819)\,\mathrm{erf}((-1)^{0.25}\,x))\,/\,\sin(x^4)$$

```
define(f_minus(x), (x · integrate(sin(t^2),t,0,x))/sin(x^4));
Is x positive, negative or zero?negative;
```

$$f\_minus(x) := 6.23666681354223\ 10^{-16}\ x\ ((251198712145819\ \%i + 251198712145819)\ \mathrm{erf}(3.126088660375976\ 10^{-8}\ (22619537\ \%i + 22619537)\ x) + (251198712145819\ \%i - 251198712145819)\ \mathrm{erf}(3.126088660375976\ 10^{-8}\ (22619537\ \%i - 22619537)\ x) + (251198712145819 - 251198712145819\ \%i)\ \mathrm{erf}((-\%i)^{0.5}\ x) + (251198712145819\ \%i + 251198712145819)\ \mathrm{erf}((-1)^{0.25}\ x))\ /\ \sin(x^4)$$

```
define(f_zero(x), (x · integrate(sin(t^2),t,0,x))/sin(x^4));
Is x positive, negative or zero?zero;
```

$$f\_zero(x) := 6.23666681354223\ 10^{-16}\ x\ ((251198712145819\ \%i + 251198712145819)\ \mathrm{erf}(3.126088660375976\ 10^{-8}\ (22619537\ \%i + 22619537)\ x) + (251198712145819\ \%i - 251198712145819)\ \mathrm{erf}(3.126088660375976\ 10^{-8}\ (22619537\ \%i - 22619537)\ x) + (251198712145819 - 251198712145819\ \%i)\ \mathrm{erf}((-\%i)^{0.5}\ x) + (251198712145819\ \%i + 251198712145819)\ \mathrm{erf}((-1)^{0.25}\ x))\ /\ \sin(x^4)$$

```
limit(f_plus(x),x,0,plus);
limit(f_minus(x),x,0,minus);
limit(f_zero(x),x,0);
```

0.0
0.0
0.0

6 Calcula una primitiva de la función 1/raiz_n(x) respecto a la variable x con n≥2

```
define(f(x),1/(x)^(1/n));
```

$$f(x) := \frac{1}{x^{1/n}}$$

integrate(f(x),x);

Is $-\frac{1}{n}$ *equal to* -1 ?no;

$$\frac{x^{1-\frac{1}{n}}}{1-\frac{1}{n}}$$

Maxima nos pregunta que si -1/n = -1. Le decimos que no porque el enunciado especifica n>=2, y
-1/n = -1;
-1 = -n;
n = 1, lo cual no es mayor o igual que 2.

INTEGRACIÓN NUMÉRICA: los ejercicios de este apartado no hay que hacerlos, pero voy a dejar aquí escritas las funciones que se usan: podemos escribir las funciones, o acceder a ellas mediante el menú Calculus>Integrate y marcando las casilla "Definite integration" y "Numerical integration". Existen dos opciones:
 -Romberg, romberg (expr, x, a, b): devuelve un número
 -Quadpack: cuenta con múltiples funciones que usará dependiendo de nuestra integral, las cuales no nos tenemos que saber. Si tenemos que saber el formato de la salida:
 > El valor de la integral
 > Cota del error
 > Número de pasos para encontrarla
 > Código de error (si es 0, la salida es correcta)

Cálculo de áreas

[redacted]R con f(x)≤g(x) en todo punto de [a,b] entonces el área del trozo de plano comprendido entre las gráficas de las funciones f y g en el intervalo [a,b] es justamente:

∫[b,a] abs(g(x) – f(x))dx.

1 Calcula el área que queda entre el eje de abscisas y la gráfica de la función f(x)=arctan(x)(x^3–x+3)*%e^(–x^2) entre el punto donde la función alcanza su mínimo relativo y en el que alcanza su máximo relativo.*

Primero vamos a calcular el máximo y el mínimo de f(x):

```
define(f(x),atan(x)·(x^3−x−3)·%e^(−x^2));
define(df(x),diff(f(x),x));
```

$$f(x):=(x^3-x-3)\,\%e^{-x^2}\,\mathrm{atan}(x)$$

$$df(x):=-2\,x\,(x^3-x-3)\,\%e^{-x^2}\,\mathrm{atan}(x)+(3\,x^2-1)\,\%e^{-x^2}\,\mathrm{atan}(x)+\frac{(x^3-x-3)\,\%e^{-x^2}}{x^2+1}$$

Graficamos la derivada para resolver con find_root:

```
wxplot2d([df(x)], [x,−5,5])$
```



Con ayuda de la gráfica determinamos los puntos en los que la derivada hace ceros:

```
find_root(df(x)=0,x,−2,0);
find_root(df(x)=0,x,0,1);
find_root(df(x)=0,x,1,2);
```

-0.6569810274741914
0.6227235852360812
1.987273577909394

Ahora, si vemos la función f(x), sabremos que podemos descartar el punto 1.98.... como máximo o mínimo relativo: los puntos a, b que buscamos son -0.6569810274741914 y 0.6227235852360812

```
wxplot2d([f(x)], [x,−5,5])$
```



Los guardamos en variables:

```
a:-0.6569810274741914;
b:0.6227235852360812;
```

$$-0.6569810274741914$$
$$0.6227235852360812$$

Pero como la función cambia de signo, tenemos que encontrar el punto c en el que se corta al eje x: así podremos obtener el área calculando primero el espacio entre el OX y la función entre a, c; y sumándole el espacio entre la función en c,b y el eje OX.

```
c:find_root(f(x),a,b);
```

$$0.0$$

Finalmente podemos calcular el área comprendida entre f(x) y el eje OX entre los puntos a,b. Seguimos la fórmula ∫[b,a] abs(g(x) – f(x)) dx, recordando que la función g(x) es la más alta; y que la función del eje OX sería obviamente y=0.

Calculamos el tramo entre a y c (f(x) por encima):

```
quad_qags(abs(f(x)), x, a, c);
```

$$\left[0.4437521167830991, 4.926638172787481\ 10^{-15}, 21, 0\right]$$

Y ahora el tramo entre c, b (eje x por encima)

```
quad_qags(abs(0-f(x)), x, c, b);
```

$$\left[0.5035139853704745, 5.590128197790751\ 10^{-15}, 21, 0\right]$$

La suma de ambos resultados es el área comprendida entre el eje x y la función:

```
area:0.4437521167830991+0.5035139853704745;
```

$$0.9472661021535737$$

2 Calcula el área que queda entre las gráficas de la función f(x)=arctan(x)*(x3-x+3)*e^(-x^2) y la función g(x)=x^3

Definimos las funciones y las graficamos. Para encontrar el intervalo a,b en el que calcular el área encontraremos los puntos en los que las funciones se cortan:

```
define(f(x),atan(x)·(x^3-x+3)·%e^(-x^2));
define(g(x),x^3);
```

$$f(x):=(x^3-x+3)\ \%e^{-x^2}\ \mathrm{atan}(x)$$

$$g(x):=x^3$$

```
wxdraw2d(
   color=blue,
   explicit(f(x), x, −4, 4),
   color=red,
   explicit(g(x), x, −4, 4),
   color=green,
   explicit(f(x)−g(x),x,−4,4),
   yrange= [−4,4],
   yaxis=true,
   xaxis=true,
   grid=true,
   proportional_axes=xy
);
```



Al igual que el ejercicio anterior, una función no es estrictamente mayor que la otra, así que calcularemos la suma de las áreas entre a,c y c,b. Vamos a calcular los puntos mencionados resolviendo su ecuación [representada con la ecuación verde, f(x)=g(x) para ver los ceros].

```
a:find_root(f(x)−g(x)=0,x,−2,−0.5);
c:find_root(f(x)−g(x)=0,x,−0.5,0.5);
b:find_root(f(x)−g(x)=0,x,0.5,2);
```

-0.9713858314297459
0.0
0.9616440225200484

Y calculamos la suma de las dos áreas: aquella en la que g(x) está por encima, y aquella en la que f(x) está por encima, ∫[b,a] abs(g(x) − f(x))dx

```
romberg(abs(g(x)−f(x)),x,a,c) + romberg(abs(f(x)−g(x)),x,c,b);
```

1.201877851875186

3 *Divide un círculo de radio 1 en tres trozos del mismo área usando dos líneas verticales, simétricas respecto del eje de ordenadas.*

Siendo la circunferencia unitaria, su área (pi*r^2) será igual a pi; podemos concluir que cada trozo tendrá un área de pi/3.

Para facilitar el ejercicio, podemos tener en cuenta sólo un cuarto de la circunferencia: en ese caso, tras dividirlo, el trozo más externo (la mitad de la circunferencia desde a hasta 1) tendrá un área de pi/3/2=pi/6, y el trozo más interno (la mitad de la circunferencia desde 0 hasta a en esta simplificación, pero en realidad es desde -a hasta a, he ahí el 4 siguiente:), tendrá un área de pi/3/4=pi/12.

Definimos directamente f(x) la integral que nos dará el área, en función del punto a:

```
define(f(x),integrate(sqrt(1−x^2),x,a,1));
```

*Is a−1 positive, negative or zero?*negative;

$$f(x):=\frac{\pi}{4}-\frac{\operatorname{asin}(a)+a\sqrt{1-a^2}}{2}$$

Al preguntarnos maxima el signo de a-1 tenemos que responder negativo, ya que a debe ser menor que 1 (perteneciendo así al interior de la circunferencia).

Para encontrar el punto a, resolvemos cualquiera de los dos trozos que hemos definido antes: hay que encontrar aquel punto en el cual f(x)=pi/6

```
numer:true$
find_root(f(x)=%pi/6,a,0,1);
```

0.2649320846027769

Los puntos en los que hay que cortar la circunferencia son 0.2649320846027769 y -0.2649320846027769.

Podemos comprobar que el área de los trozos es de pi/3:

```
numer:true$
ratprint: false$
A1:integrate(2·sqrt(1−x^2),x,0.2649320846027768,1);
A2:integrate(2·sqrt(1−x^2),x,−0.2649320846027768,0.2649320846027769);
A2:integrate(2·sqrt(1−x^2),x,−1,−0.2649320846027768);
```

1.047197551196598
1.047197551196598
1.047197551196598

```
float(%pi/3);
```

1.047197551196598

Las tres áreas miden un tercio.

4 Calcula las dos áreas en los que la función f(x)=|x|–x*sin(x)*e^x divide al círculo x^2+y^2=1

```
define(f(x),abs(x)−x·sin(x)·%e^x);
define(g(x),sqrt(1−x²));
```

$$f(x):=|x|-x\,\%e^{x}\,\sin(x)$$

$$g(x):=(1-x^{2})^{0.5}$$

```
wxdraw2d(
   color=blue,
   explicit(f(x), x, −2, 2),
   color=red,
   implicit(x²+y²=1, x, −2, 2,y,−2,2),
   color=green,
   explicit(f(x)−g(x),x,−2,2),
    color=orange,
   explicit(−f(x)−g(x),x,−2,2),
   yrange= [−2,2],
   yaxis=true,
   xaxis=true,
   grid=true,
   proportional_axes=xy
);
```



Antes que nada, vamos a calcular y marcar los puntos de corte de las dos funciones. Las curvas naranja y verde son la representación de f(x)-g, para poder delimitar el rango a usar en find_root.

```
a:find_root(f(x)−g(x)=0,x,−1,0);
b:find_root(−f(x)−g(x)=0,x,0,1);
```

$$-0.8276331360743349$$

$$0.8265203122224214$$

```
wxdraw2d(
    color=blue,
    explicit(f(x), x, −2, 2),
    color=red,
    implicit(x²+y²=1, x, −2, 2,y,−2,2),
    color=green,
    implicit(x=a+y·0,x,−2,2,y,−2,2),
    color=orange,
    implicit(x=b+y·0,x,−2,2,y,−2,2),
    yrange= [−2,2],
    yaxis=true,
    xaxis=true,
    grid=true,
    proportional_axes=xy
);
```



He ilustrado los puntos de corte con líneas rectas para más claridad. Para calcular las dos áreas procedemos de la siguiente forma:

- Calculamos el área entre de la curva azul y el círculo (mitad inferior), entre los puntos de corte a,b
- Pero con eso se nos queda fuera el área comprendida ("casquete") entre el círculo y la recta verde.

Para calcularla, calculamos el área entre el círculo y el eje OX entre -1 y el punto a. La multiplicamos por dos (porque no hemos tenido en cuenta la parte debajo del eje OX).

-Repetimos el procedimiento pero en la mitad superior del círculo, incluyendo el casquete delimitado por la línea naranja.

```
A_inf: romberg(abs(f(x)−g(x)),x,a,b) + 2·romberg(abs(g(x)),x,−1,a);
```

    1.311635504916393

```
A_sup: abs(romberg(abs(g(x)−f(x)),x,a,b) + 2·romberg(abs(g(x)),x,1,a));
```

    1.829912399102185

Finalmente, podemos comprobar que es correcto conociendo que el área del círculo es pi*r^2, y dado que es la circunferencia unitaria, el área debería ser igual a pi:

```
A_sup+A_inf;
```

    3.141547904018578

5 *Calcula el área de la figura limitada por la curva y=x^3−x^2 y el eje de abscisas.*

```
define(f(x),x^3−x^2);
```

$$f(x):=x^3-x^2$$

```
wxplot2d([f(x)], [x,−500,500])$
```



No tenemos con qué delimitar el área de la figura: como la función tiende a -∞ por la izquierda y a ∞ por la derecha, el área es infinita. Maxima nos dará un mensaje de error ante cualquier intento de resolver las integrales entre -∞,0 o 0,∞.

Longitudes de curvas

[redacted] [a,b] y queremos calcular la longitud de la gráfica L, la fórmula es
L = ∫[ba] sqrt(1+ (df(x))^2) dx.

1 *Calcula la longitud del arco de curva y=x^2+4 entre x=0 y x=3*

```
define(f(x),x^2+4);
define(df(x),diff(f(x),x));
```

$$f(x):=x^2+4$$

$$df(x):=2\ x$$

```
integrate(sqrt(1+df(x)^2),x,0,3);
```

9.747088758608557

2 *Calcula la longitud de una elipse de semiejes 10 y 15*

La ecuación de una elipse centrada en el origen es x^2/a^2+y^2/b^2=1. Si queremos calcular la longitud, podemos calcular la longitud de la semielipse positiva y multiplicarla por dos: y=sqrt((1-x^2/a^2)*b^2)

```
define(f(x),sqrt((1−x^2/10^2)·15^2));
define(df(x),diff(f(x),x));
```

$$f(x):=15.0\,(1-0.01\,x^2)^{0.5}$$

$$df(x):=-\frac{0.15\,x}{(1-0.01\,x^2)^{0.5}}$$

```
quad_qags(sqrt(1+df(x)^2), x, −10, 10);
```

$$[39.66359897324617, 7.926999501250975\,10^{-9}, 567, 0]$$

```
2·39.66359897324617;
```

$$79.32719794649233$$

3 Calcula la longitud del arco de la curva 9*x^2=4*y^3 entre los puntos (0,0) y (2*sqrt(3),3)

Primero despejamos la función: y=((9*x^2)/4))^(1/3)

```
numer:false$
define(f(x),(((9/4)·x^2)/4)^(1/3));
define(df(x),diff(f(x),x));
```

$$f(x):=\frac{9^{1/3}\,x^{2/3}}{2^{4/3}}$$

$$df(x):=\frac{9^{1/3}}{3\,2^{1/3}\,x^{1/3}}$$

```
quad_qags(sqrt(1+df(x)^2), x, 0, 2·sqrt(3));
```

$$[4.007053671189008, 5.943146419440382\,10^{-9}, 315, 0]$$

4 La gráfica de la función y=(x^3)+1/e divide el círculo centrado en el origen de radio 3 en dos partes. Calcula la longitud de la gráfica que queda dentro de la circunferencia.

```
define(f(x),x^3+1/%e);
define(df(x),diff(f(x),x));
```

$$f(x):=x^3+\%e^{-1}$$

$$df(x):=3\,x^2$$

```
g(x):=sqrt(3²−x^2);
```

$$g(x) := \sqrt{3^2 - x^2}$$

```
wxdraw2d(
   color=blue,
   explicit(f(x), x, −4, 4),
   color=red,
   implicit(x²+y²=3², x, −4, 4,y,−4,4),
   color=green,
   explicit(f(x)−g(x), x, −4, 4),
   color=orange,
   explicit(−f(x)−g(x), x, −4, 4),
   yrange= [−4,4],
   yaxis=true,
   xaxis=true,
   grid=true,
   proportional_axes=xy
);
```



Tenemos que calcular los puntos de corte de la función con la circunferencia. En la gráfica nos ayudamos con las curvas verde y naranja para saber donde aplicar find_root.

```
find_root(−f(x)−g(x)=0,x,−2,−1);
find_root(f(x)−g(x)=0,x,1,2);
```

$$-1.442021620435344$$
$$1.324557969592638$$

Podemos comprobar que esos son los puntos de corte:

```
wxdraw2d(
   color=blue,
   explicit(f(x), x, −4, 4),
   color=red,
   implicit(x²+y²=3², x, −4, 4,y,−4,4),
   color=black,
   point_type=7,
   point_size=2,
   points([
               [−1.442021620435344, f(−1.442021620435344)],
               [1.324557969592638,   f(1.324557969592638)]
            ]),
   yrange= [−4,4],
   yaxis=true,
   xaxis=true,
   grid=true,
   proportional_axes=xy
);
```



Así que aplicamos la fórmula:

```
quad_qags(sqrt(1+df(x)^2), x, −1.442021620435344, 1.324557969592638);
```

$$\left[6.508686583034771, 2.635469839199217\ 10^{-8}, 105, 0\right]$$

Volúmenes de sólidos de [redacted]

[redacted] f:[a,b]→R el volumen del sólido de revolución obtenido al girar alrededor del eje X el área que queda entre la gráfica de la función f y dicho eje viene dado por:

V=π*∫[b,a] f(x)^2 dx

Si f es una función positiva en un intervalo [a,b] de números positivos, el volumen del objeto obtenido al girar respecto del eje Y es:

V=2π*∫[b,a] x*f(x) dx.

1 Calcula el volumen del sólido generado al girar la gráfica de la curva y=sin(x)^2, para x∈[0,π], alrededor del eje OX.

```
quad_qags(sin(x)^2, x, 0, %pi);
```

$$\left[1.570796326794897, 1.743934249004316\ 10^{-14}, 21, 0\right]$$

```
float(%pi·1.570796326794897);
```

4.93480220054468

2 Calcular el volumen del sólido engendrado al girar alrededor del eje OY la región limitada por la parábolas y^2=x, x^2=y

Primero declaramos las funciones en función de x:

```
define(f(x),sqrt(x));
define(g(x),x^2);
```

$$f(x):=\sqrt{x}$$

$$g(x):=x^2$$

```
wxdraw2d(
  color=blue,
  explicit(g(x), x, −4, 4),
  color=red,
  implicit(y^2=x, x, −4, 4,y,−4,4),
  /*color=green,
  explicit(f(x)-g(x), x, -4, 4),
  color=orange,
  explicit(-f(x)-g(x), x, -4, 4),*/
  yrange= [−4,4],
  yaxis=true,
  xaxis=true,
  grid=true,
  proportional_axes=xy
);
```



Es decir, nos están pidiendo el volumen de la figura que generaría ese espacio entre las dos funciones en forma de "lágrima" al girarlo en el eje Y. Quedaría una especie de cono invertido (sin tapa)

```
quad_qags(x·(sqrt(x)−x^2), x, 0, 1);
```

$$[0.1500000000000001, 1.665334536937736\ 10^{-16}, 189, 0]$$

```
float(2·%pi·0.1500000000000001);
```

$$0.9424777960769386$$

3 Sea f(x)=x^5+4*x^3+2*x^2+8. Calcula el volumen al girar dicha función alrededor del eje OX entre los valores donde f alcanza su máximo y su mínimo relativos.

Primero tenemos que definir la función para calcular su máximo y mínimo relativos:

```
define(f(x),x^5+4·x^3+2·x^2+8);
define(df(x),diff(f(x),x));
```

$$f(x):=x^5+4x^3+2x^2+8$$

$$df(x):=5x^4+12x^2+4x$$

```
solve(df(x)=0,x);
```

$$[x=\left(\frac{2\sqrt{21}}{5^{3/2}}-\frac{2}{5}\right)^{1/3}\left(\frac{-1}{2}-\frac{\sqrt{3}\,\%i}{2}\right)-\frac{4\left(\frac{\sqrt{3}\,\%i}{2}+\frac{-1}{2}\right)}{5\left(\frac{2\sqrt{21}}{5^{3/2}}-\frac{2}{5}\right)^{1/3}}, x=\left(\frac{2\sqrt{21}}{5^{3/2}}-\frac{2}{5}\right)^{1/3}\left(\frac{\sqrt{3}\,\%i}{2}+\frac{-1}{2}\right)-\frac{4\left(\frac{-1}{2}-\frac{\sqrt{3}\,\%i}{2}\right)}{5\left(\frac{2\sqrt{21}}{5^{3/2}}-\frac{2}{5}\right)^{1/3}}, x=\left(\frac{2\sqrt{21}}{5^{3/2}}-\frac{2}{5}\right)^{1/3}-\frac{4}{5\left(\frac{2\sqrt{21}}{5^{3/2}}-\frac{2}{5}\right)^{1/3}}, x=0]$$

De las soluciones obtenidas podemos descartar las imaginarias, quedándonos 2:

```
a:((2·sqrt(21))/5^(3/2)−2/5)^(1/3)−4/(5·((2·sqrt(21))/5^(3/2)−2/5)^(1/3));
```

$$\left(\frac{2\sqrt{21}}{5^{3/2}}-\frac{2}{5}\right)^{1/3}-\frac{4}{5\left(\frac{2\sqrt{21}}{5^{3/2}}-\frac{2}{5}\right)^{1/3}}$$

```
b:0;
```

$$0$$

```
wxdraw2d(
   color=blue,
   explicit(f(x), x, −3, 3),
   color=black,
   point_type=7,
   point_size=2,
   points([
            [a, f(a)],
            [b, f(b)]
         ]),
   yrange= [−2,10],
   yaxis=true,
   xaxis=true,
   grid=true
);
```



```
quad_qags(f(x)^2, x, a, b);
```

$$\left[20.64099418331932, 2.291610699347504\ 10^{-13}, 21, 0\right]$$

```
float(%pi·20.64099418331932);
```

$$64.84559568910564$$

4 *Halla mediante integración el volumen de un cono circular recto de altura h y con base una circunferencia de radio r.*

Un cono es una figura generada por una línea recta que parte del origen de coordenadas a un punto (h,r) alrededor del eje OX.

La ecuación de una recta que pasa por dos puntos es y-y1= m*(x-x1), donde m=(y2-y1)/(x2-x1). Siendo el punto 1 (0,0) y el punto 2 (h,r):
y=(h/r)*x

```
define(f(x),(r/h)·x);
```

$$f(x):=\frac{r\,x}{h}$$

```
V:%pi·integrate(f(x)^2,x,0,h);
```

$$\frac{\pi\,h\,r^{2}}{3}$$

Áreas de sólidos de [redacted]

[redacted] una superficie de revolución que se obtiene al girar alrededor del eje x la gráfica de una función f positiva en un intervalo [a,b]:

2π*∫[b,a] f(x)*sqrt(1+df(x)^2) dx

1 *Se suele llamar "La trompeta de Gabriel" a la figura que se obtiene al girar alrededor del eje de abscisas la gráfica de la función f(x)=1/x entre 1 e infinito. Calcula el valor del volumen que encierra y la superficie de dicha figura.*

```
define(f(x),1/x);
define(df(x),diff(f(x),x));
```

$$f(x):=\frac{1}{x}$$

$$df(x):=-\frac{1}{x^{2}}$$

```
V:%pi·integrate(f(x)^2,x,1,inf);
```

$$\pi$$

```
A:2·%pi·integrate(f(x)·sqrt(1+df(x)^2),x,1,inf);
```

defint: integral is divergent.
 -- an error. To debug this try: debugmode(true);

El volumen que encierra es igual a pi, pero su área es infinita.

2 Calcula, usando integración, cuanto mide la superficie de una bola de radio r.

Podemos tomar la esfera como un sólido de revolución resultante de rotar la semicircunferencia superior en el eje X

```
define(f(x),sqrt(r²−x²));
define(df(x),diff(f(x),x));
```

$$f(x):=\sqrt{r^2-x^2}$$

$$df(x):=-\frac{x}{\sqrt{r^2-x^2}}$$

```
A: 2·π·integrate(f(x)·sqrt(1+df(x)^2),x);
```

$$2\,\pi\,r\,x$$

3 Halla mediante integración el área lateral de un cono circular recto de altura h y con base una circunferencia de radio r.

Un cono es una figura generada por una línea recta que parte del origen de coordenadas a un punto (h,r) alrededor del eje OX.

```
define(f(x),(r/h)·x);
define(df(x),diff(f(x),x));
```

$$f(x):=\frac{r\,x}{h}$$

$$df(x):=\frac{r}{h}$$

```
A: 2·π·integrate(f(x)·sqrt(1+df(x)^2),x);
```

$$\frac{\pi\,r\sqrt{\frac{r^2}{h^2}+1}\;x^2}{h}$$